/*-----------------------------------------------------------------------------
	The contents of this file are subject to the Mozilla Public License
	Version 1.1 (the "License"); you may not use this file except in
	compliance with the License. You may obtain a copy of the License at
	http://www.mozilla.org/MPL/
	
	Software distributed under the License is distributed on an "AS IS"
	basis, WITHOUT WARRANTY OF ANY KIND, either express or implied. See the
	License for the specific language governing rights and limitations
	under the License.
	
	The Original Code is Fever Framework code.
	
	The Initial Developer of the Original Code is Romain Ecarnot.
	Portions created by Initial Developer are Copyright (C) 2006
	the Initial Developer. All Rights Reserved.
	
	Contributor(s):
	
	Unless required by applicable law or agreed to in writing, software
	distributed under the License is distributed on an "AS IS" BASIS,
	WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
	See the License for the specific language governing permissions and
	limitations under the License.
-----------------------------------------------------------------------------*/

import com.bourre.data.collections.Map;
import com.bourre.data.iterator.Iterator;
import com.bourre.utils.ClassUtils;

import fvaswing.components.FvUIBuffer;

/**
 * Allow to insert component into special buffer to preload them before any 
 * actions in your application.
 * 
 * <p>Buffer is preload during {@code fvaswing.FvAsWing} instanciation.<br />
 * Adds components to buffer before {@code FvAsWing.getInstance()} or 
 * {code FvAsWing.init()} methods call.
 * 
 * <p>{@link fever.display.FeverStage} broadcast {@link fvaswing.FvAsWing#onUICompleteEVENT} 
 * event when components are preloaded and UI is ready to be used. ( just after the 
 * 1st UI paint ).
 *
 * @author Romain Ecarnot
 */
class fvaswing.utils.FvComponentBuffer
{
	//-------------------------------------------------------------------------
	// Private properties
	//-------------------------------------------------------------------------
	
	private static var _buffer : Map = new Map();
	
	
	//-------------------------------------------------------------------------
	// Public API
	//-------------------------------------------------------------------------
	
	/**
	 * Adds passed-in component into buffer.
	 */
	public static function push( component : FvUIBuffer ) : Void
	{
		var className : String = ClassUtils.getFullyQualifiedClassName( component );
		
		if( !_buffer.containsKey( className ) )
		{
			_buffer.put( className, component );
		}
	}
	
	/**
	 * Returns {@code true} if no components are added to buffer.
	 */
	public static function isEmpty(  ) : Boolean
	{
		return _buffer.isEmpty();
	}
	
	/**
	 * Starts preloading of all included components.
	 */
	public static function preload() : Void
	{
		var it : Iterator = _buffer.getValuesIterator();
		
		while( it.hasNext() ) it.next().preload();
		
		_buffer.clear();
	}
}